

Using SAS® to Support the Implementation of a Patient-Centered Outcomes Research Institute Grant Funded by the Affordable Care Act

Michele Mesiano, Community Care Behavioral Health Organization; Meghna Parthasarathy, Community Care Behavioral Health Organization; Lauren Terhorst, University of Pittsburgh

ABSTRACT

The Patient-Centered Outcomes Research Institute (PCORI) was created as part of the Affordable Care Act. PCORI is authorized by Congress to conduct research to provide information about the best available evidence to help patients and their health care providers make more informed decisions. Community Care Behavioral Health Organization in Pittsburgh, Pennsylvania was awarded a PCORI research grant to investigate health care system improvements for adults with serious mental illness. The grant, titled “Optimizing Behavioral Health Homes by Focusing on Outcomes that Matter Most for Adults with Serious Mental Illness,” began in January of 2013 and is ongoing. Information Technology staff at Community Care have leveraged SAS® solutions in providing real-time data extraction and reports to support the development and implementation of this research project. SAS tools have been used to merge data from multiple platforms and database sources, including web data sources. SAS has also enabled the formatting and traffic lighting of multiple Microsoft Excel data sets and files, in addition to the creation of many operational reports and data files needed for study implementation, administration, and maintenance. The challenges faced and the SAS solutions employed are the subject of this paper.

INTRODUCTION

The Patient-Centered Outcomes Research Institute was created as part of the 2010 Patient Protection and Affordable Care Act (Public Law 111-148).¹ The section of the law that creates PCORI amends Title XI of the Social Security Act. The law authorizes the establishment of a nonprofit corporation that is not part of the federal government that is charged with examining the health outcomes, clinical effectiveness, and appropriateness of different medical treatments by evaluating existing studies and conducting its own. The purpose of the Institute is to assist patients, clinicians, purchasers, and policy-makers in making informed health decisions by advancing the quality and relevance of research evidence on how health conditions can be prevented, diagnosed, treated and managed. It is also charged with establishing ways to disseminate these research findings into the general population.

Simply stated, the mission of PCORI is to “help people make informed healthcare decisions, and improve healthcare delivery and outcomes, by producing and promoting high integrity, evidence-based information that comes from research guided by patients, caregivers and the broader healthcare community”.² Improved health care models are imperative because the United States spends more on health care than any other country, yet our life expectancy is lower than other developed countries.

Community Care Behavioral Health Organization (Community Care) is a member of the Insurance Services Division of the University of Pittsburgh Medical Center (UPMC). Community Care is the only Pennsylvania-based, not-for-profit, licensed behavioral health managed care organization (BH-MCO). Community Care was created to serve HealthChoices programs throughout Pennsylvania. Community Care holds both risk-bearing and Administrative Services Only (ASO) contracts. The organization has eight offices across the Commonwealth of Pennsylvania and currently has contractual arrangements with 39 of the 67 Pennsylvania counties. With nearly 500 employees and by utilizing a network of approximately 2,000 behavioral health providers, Community Care manages behavioral healthcare for over 700,000 Medicaid managed care enrollees. Community Care was designed primarily to serve the needs of public sector consumers, their families, and their communities. The mission of Community Care is to improve the health and well-being of the community through the delivery of clinically effective, cost-efficient, and accessible behavioral health services. In December of 2012, Community Care was awarded a PCORI grant based upon their application entitled “Optimizing Behavioral Health Homes by Focusing on Outcomes that Matter Most for Adults with Serious Mental Illness”.

Persons suffering from serious mental illness, a diagnosable mental health disorder defined by duration and functional impairment, are some of the most medically vulnerable populations in America. It is estimated that 6 percent (or 1 in 17) adults live with a serious mental illness.³ These people not only struggle with their behavioral health condition; they often have high physical health need and difficulty accessing physical health care. Difficulty accessing care is often higher in rural areas. Physical health problems in this cohort cover a broad spectrum, including cardiovascular disease, diabetes, hypertension, obesity, and hyperlipidemia. As a result, adults with serious mental illness have high rates of premature death, dying as much as 15 to 25 years younger than the general

population.⁴

The grant, “Optimizing Behavioral Health Homes by Focusing on Outcomes that Matter Most for Adults with Serious Mental Illness”, provides a unique opportunity to study the effectiveness of wellness interventions across 11 Community Mental Health Centers in the north central and eastern regions of Pennsylvania. The research tests the comparison between two promising practices for promoting health, wellness, and recovery of adults with serious mental illness: patient self-directed care and provider-supported integrated care. Patient self-directed care is practiced at 6 of the provider locations participating in the research study. This arm of treatment focuses on activating patients to be more informed and effective managers of their own health and health care. It uses a specialized web portal to deliver health information to patients as well as tools for tracking their progress on their personal wellness goals. Additionally, a specially-trained peer with similar life issues and experiences is provided to the patient to assist them in taking a more active role in their health and health care. Provider-supported integrated care is practiced at 5 of the provider locations participating in the research study. This arm of treatment utilizes a registered nurse on staff at the locations who works with patients to provide patient wellness support and education, as well as to assist with coordination of care and to monitor the patients’ medical progress. Both arms of treatment focus on health and wellness education, the setting of wellness goals, creating action plans to meet those goals, and monitoring progress using health trackers. The specific areas of wellness focused on are: healthy weight, smoking cessation, physical activity, taking medications effectively, physical and behavioral health care, sleep hygiene, and stress reduction.

The implementation of this grant provided many technical challenges to Community Care. Data with disparate formats from multiple sources and platforms had to be combined to meet the many data collection, extraction and reporting needs of the project. Member data was comprised of data obtained from paid Medicaid behavioral service claims, paid Medicaid pharmacy claims, Medicaid service authorization data, member reported data and web portal data from participating providers. Data was pulled to identify and stratify members for the project; data was then sent to and used by the providers to begin study recruitment. A web portal was developed to collect member consent and to provide a data entry tool for member and provider measurement, survey, and other study related data. Web-based health trackers were also made available to study participants. Multiple complex reports were needed for study implementation and administration. Due to the disparity in data sources and formats as well as the demanding requirements of the PCORI project, SAS[®] was determined to be the most efficient tool for data manipulation.

PROJECT IMPLEMENTATION

All member information was extracted from a behavioral health claims database using SAS[®]. Study eligible members were identified using the following criteria: a) 18 years of age or older; b) member of Medicaid Health Choices in the North Central area of Pennsylvania; and c) a diagnosis of schizophrenia, major mood disorder, psychotic disorder, or borderline personality disorder. Additionally, members were identified by number of claims for specific types of behavioral health service utilization within a 6 month look back period. Members qualified for study enrollment if 2 or more claims were found documenting use of case management, certified peer specialist contacts, Assertive Community Treatment (ACT) team contacts, psychiatric rehabilitation services or outpatient services within the 6 month period. Next, the provider sites at which the members received services were listed to provide a nested data structure. If a member received differing types of services at different provider sites, the site providing case management services became the primary site.

Once the list of members who qualified for enrollment into PCORI at the various provider sites was compiled, they were stratified into a high and low risk tier. The purpose of these tiers is to help focus efforts towards the members who had the highest health risk factors in both treatment arms of this study in case of staffing or administrative shortfalls.

The high risk tier comprised of members who had chronic physical health conditions like diabetes or asthma or Chronic Obstructive Pulmonary Disease (COPD) (identified by pharmacy insurance claims for medications filled for any of these conditions) or had received high intensity behavioral health services in the prior year such as multiple inpatient hospitalizations, admission into an extended acute care unit, multiple services with Assertive Community Treatment (ACT) Teams, multiple services with Community Treatment Teams (CTT), or had concomitant fills for multiple behavioral health medications such as antipsychotics, antidepressants, mood stabilizers, opiates, anxiolytics, stimulants, sedatives, etc. Approximately 30% of the members who qualified for recruitment into PCORI were in the high risk tier.

After stratification, member lists were generated for each provider using the ODS HTML utility in SAS in combination with PROC REPORT (Display 1 and Display 2). The generated excel files were then password protected per HIPAA

regulations since the files contained protected health information and were distributed to the participating providers via company secure email.

```
%macro prov_output(indata) ;

ODS LISTING CLOSE;

proc sort data = &indata
out=outdata1 ;
by provider_name ;
run;

data outdata2 ;
set outdata1 ;
by provider_name;
retain prov_index 0;
if first.provider_name then prov_index+1;
run;

proc sql noprint ;
select max(prov_index) into: maxind
from outdata2
where prov_index > 0;
quit;

%do i = 1 %to &maxind ;

proc sql noprint ;
select distinct strip(compress(provider_name, " _-")) into :provname
from outdata2
where prov_index=&i;
quit;

data prov_&i_data;
set outdata2 ;
where prov_index=&i;
run;

proc sql noprint ;
select distinct provider_name into :prov
from prov_&i_data ;
quit ;
```

Using a SAS macro and a DO LOOP, separate datasets are created with all the relevant information for each of the 11 participating providers.

Display 1: Macro and the DO Loop

```
ods html body = "\\Client\C$\Users\parthasarathym\Documents\PCORI_Share\&provname_&sysdate9.xls"
style = egdefault ;

Title "MEMBER REPORT FOR PCORI IDENTIFICATION";
Title2 "PROVIDER NAME: &prov ";
Title3 "File date: &sysdate9 ";
Title4 " ";
Title5 " ";
Title6 " ";
footnote height=10pt "Community Care Behavioral Health | Department:Decision Support & Data Analytics" ;

proc report data = prov_&i._data headline headskip split="*" nowd style(report)={cellpadding=3pt} ;
column provider_name provider_id insured_id tier member_name dob consentTF consentdate ;

define provider_name / noprint ;
define provider_id / noprint ;

define insured_id / display "Medicaid ID"
style={htmlstyle="mso-number-format:'0000000000'"
cellwidth=15% font_face=calibri font_size=3 } ;

define member_name / display "Member Name"
style={cellwidth=15% font_face=calibri font_size=3 } ;

define dob / display "DOB"
style={cellwidth=10% just=center font_face=calibri font_size=3 } format=date11. ;

define tier / display "RISK TIER "
style={cellwidth=10% just=center font_face=calibri font_size=3 } ;
define consentTF / display "PCORI CONSENT (YES=1, NO=0) "
style={cellwidth=15% just=center font_face=calibri font_size=3 } ;

define consentdate / display "PCORI Consent Date"
style={cellwidth=10% just=center font_face=calibri font_size=3 } format=date11. ;

compute before _page_ / Left
style={just=left font_face=calibri font_size=3 font_weight=bold} ;
Line "Provider Name: " provider_name $48. ;

endcomp;
run;
%end ;
ODS html CLOSE;
ODS LISTING;
%mend ;

%prov_output(DAT.ID_STRAT_LIST_&SYSDATE9) ;
```

ODS HTML is used to create a separate excel document for each provider.

The ability to link an XML format to the excel file in PROC REPORT was most useful here because a member's Medicaid ID is always 10 digits and therefore has to be backfilled with zeros.

Without the XML formatting, the leading zeros would always be dropped.

Display 2: ODS HTML and PROC REPORT to Create Multiple Excel Files

CONSENT PROCESS

Once the provider sites received their file, they used it to guide and inform their PCORI recruitment. The providers were also allowed to recruit members for the PCORI study who were not on the identified list sent to them as long as the member met all the identification criteria. At each provider site, interested members logged on to a web portal built specifically for this project and submitted their consent for this study.

In addition to the consent, members use the web portal to complete a set of seven survey instruments which include the Physical Health Information (PH Part 1) and the Interpersonal Support Evaluation List (ISEL) which help assess the member's quality of life in personal and professional settings while living with serious chronic behavioral and physical health conditions (shown in Display 3).

Answered in a pre-determined sequence, the member's responses to these surveys are collected at least four times during the study. The member screen and the order in which the surveys appear are shown in Display 3. The first time they complete the survey is recorded as a baseline response. Subsequent surveys are re-measurements at different time points in the study. These responses will be used to measure outcomes and assess the effectiveness of the different treatment arms at the end of the project.

The consent data and survey response data feeds into a Cognos framework developed for this purpose on a daily basis and is then extracted into an excel file using the Cognos Report Query created for this project.

Display 3: Member Web Portal for PCORI Survey Response Collection

Each tab in the excel file contained the responses from one form on the web, i.e. the data from the consent form was in the first tab, data from the SF12 was in the second tab and similarly each of the others surveys shown in Display 3 were in a separate tab as well. A screen shot of the excel file is shown in Display 4.

	A	B	C	D	E
	AccessNumber	ConsentTf	Age	ConsentDate	FACILITY_CODE
1		1	57	Sep 19, 2013 3:10:15 PM	1000961
2		1	53	Sep 20, 2013 9:02:32 AM	1000491
3		1	61	Sep 20, 2013 10:42:32 AM	1000491
4		1	25	Sep 20, 2013 12:26:51 PM	1000961
5		1	60	Sep 20, 2013 12:37:03 PM	1000961
6		1	40	Sep 20, 2013 1:48:23 PM	1000491
7		1	58	Dec 12, 2013 1:59:09 PM	1000701
8		1	50	Nov 26, 2013 4:04:16 PM	1000701
9		1	51	Nov 26, 2013 6:03:09 PM	1000701
10		1	41	Nov 27, 2013 9:12:35 AM	1001046
11		1	44	Nov 15, 2013 10:30:08 AM	1000811
12		1	52	Nov 12, 2013 2:07:28 PM	1000191
13		1	26	Dec 20, 2013 1:04:11 PM	1000491
14		1	61	Nov 15, 2013 1:30:05 PM	1000811
15		1	23	Nov 18, 2013 3:44:58 PM	1000701
16		1	75	Nov 21, 2013 11:07:24 AM	1000811
17		1	42	Nov 24, 2013 12:00:00 AM	1000468
18		1	28	Dec 17, 2013 11:27:27 AM	1000961
19		1	25	Nov 6, 2013 11:58:37 AM	1001046
20		1	49	Dec 6, 2013 10:57:58 AM	1000961
21		1	33	Oct 18, 2013 12:00:00 AM	1001046
22		1	36	Nov 25, 2013 11:47:58 AM	1001046
23		1	44	Dec 12, 2013 11:26:22 AM	1000191
24		1	48	Nov 22, 2013 12:21:05 PM	1000701
25		1	-1	Nov 20, 2013 3:00:03 PM	1000701
26		1	51	Nov 11, 2013 9:53:20 AM	1000191
27		1	60	Oct 21, 2013 13:00:00 AM	1000961

Display 4: Excel File Created from Cognos

Excel data in the format shown in Display 4 enabled the use of the code shown in Display 5. The result of using this is shown in Display 6.

```
libname pcoriweb odbc required=
  "Driver={Microsoft Excel Driver (*.xls, *.xlsx, *.xlsm, *.xlsb)};
  DBQ=\\managed-psych12\Analysis\Development Area\Meghna's Development Folder\
  PCORI Grant\Go Live Reports\Web files\WEB_02142014.xlsx;";
run;

**Generate a list of available excel data tabs **;
ods output Members=dnames;
proc datasets library=pcoriweb memtype=data;
quit;
run;

ods output close;
run;


**Strip the dollar-signs($) from the dataset names. (SAS puts '$' in the dataset names.
Get rid of those for use in SAS data steps ...) Also keep only 'SYSTEM TABLE' names **;
data dnames2(keep=dataname name);
set dnames;
dataname=compress(name,'$');
if dbmstype='SYSTEM TABLE';
run;

** This macro/call-execute runs through the dataset list and
'moves' the excel data to SAS work library dataset members **;
data _null_;
set dnames2;
mySAS="data " || trim(dataname) || ";set pcoriweb." || trim(name) || "n;run;";
call execute(mySAS);
run;
```

The libname statement creates a libname directly to the excel file and each of the sheets is now a dataset with a \$ sign appended to the sheet name that is present in excel. The first row in each tab is now the variable name. The result is shown in Display 6.

The consent table is shown in Display 7 as an example.

Display 5: SAS Code Importing Excel Data Directly into SAS



Display 6: Result of the Libname to the Excel File

AccessNumber	ConsentTF	Age	ConsentDate	FACILITY_CODE
	1	57	19SEP2013 15:10:14	1000961
	1	53	20SEP2013 09:02:31	1000491
	1	61	20SEP2013 10:42:31	1000491
	1	25	20SEP2013 12:26:50	1000961
	1	60	20SEP2013 12:37:02	1000961
	1	40	20SEP2013 13:48:22	1000491
	1	58	12DEC2013 13:59:08	1000701
	1	50	26NOV2013 16:04:16	1000701
	1	51	26NOV2013 18:03:09	1000701
	1	41	27NOV2013 09:12:34	1001046
	1	44	15NOV2013 10:30:07	1000911
	1	52	12NOV2013 14:07:27	1000191
	1	26	20DEC2013 13:04:10	1000491
	1	61	15NOV2013 13:30:05	1000911

Display 7: Consent Table as Sample

Each of the tabs in the excel file were therefore converted to permanent SAS datasets with minimal manual intervention.

PROVIDER PERFORMANCE TRACKING

There were three rates that needed tracked for the initial recruitment phase of the project – the consent rate, primary measure completion rate and survey completion rate.

The consent rate corresponded to the rate of members who qualified for the PCORI study that consented to be part of the study. The primary measure completion rate was the rate of members who consented for the PCORI study and also completed the first three surveys of the study shown in Display 3. This rate was important because these three surveys are part of the primary outcomes to be studied at the end of this PCORI grant. The last tracking rate was the survey completion rate which was the rate of members who consented for the PCORI study and completed all seven of the surveys.

All rates were presented as an overall rate, a rate by treatment arm and a rate by provider. A combination of PROC SQL, the DATA STEP MERGE and PROC REPORT was used to create the final output.

Member consents and member responses to the PH Part 1 survey (third survey in Display 3) and ISEL survey (seventh survey in Display 3), all of which were converted to permanent SAS datasets as shown in Display 7 were grouped individually to create summary datasets for PCORI study consent, the PH Part 1 survey and the ISEL survey - by provider, by treatment arm and overall for the entire project.

These summary datasets were then merged to the summary dataset created from the original member identification file sent to all providers which showed the distribution of members who qualified for the PCORI study – by provider, treatment arm and overall. The next few figures explain the building of the final output table and the final report.

```

174 data all2 ;
175 merge prov_freq2 (in=a) prov_consent_freq2 (in=b) prov_isel16_freq2 (in=c)
176 prov_isel16_freq3 (in=d) ;
177 by provider_name provider_id ;
178 if a ;
179 format pct_yes percent13.2 pct_complete percent13.2 pct_primary percent13.2 ;
180 if consentTF = 1 then pct_yes = consent_ID/all_ID ;
181 if isel16TF = 1 and consentTF = 1 then pct_complete = isel16_ID / consent_ID ;
182 if phiTF = 1 and consentTF = 1 then pct_primary = phi1_ID / consent_ID ;
183 run;

```

consentTF = 1 refers to members who consented to the PCORI study, isel16TF = 1 refers to members who finished the last survey shown in Display 3 which indicates that they have finished all the seven required surveys, PHITF = 1 refers to members who finished the first 3 surveys shown in Display 3.

Display 8: Merging Summary Datasets

The dataset created from the above merge is shown in Display 9.

VIEWTABLE: Work.All2											
	provider_name	all_ID	ConsentTF	consent_ID	isel16TF	isel16_ID	phiTF	phi1_ID	pct_yes	pct_complete	pct_primary
1	Provider 1	120	1	54	1	54	1	54	45.00%	100.00%	100.00%
2	Provider 2	176	1	82	1	79	1	80	46.59%	96.34%	97.56%
3	Provider 3	140	1	83	1	83	1	83	59.29%	100.00%	100.00%
4	Provider 4	231	1	113	1	113	1	113	48.92%	100.00%	100.00%
5	Provider 5	74	1	27	1	27	1	27	36.49%	100.00%	100.00%
6	Provider 6	96	1	62	1	61	1	61	64.58%	98.39%	98.39%
7	Provider 7	102	1	28	1	28	1	28	27.45%	100.00%	100.00%
8	Provider 8	445	1	306	1	306	1	306	68.76%	100.00%	100.00%
9	Provider 9	133	1	99	1	99	1	99	74.44%	100.00%	100.00%
10	Provider 10	205	1	175	1	174	1	175	85.37%	99.43%	100.00%
11	Provider 11	126	1	70	1	70	1	70	55.56%	100.00%	100.00%

Display 9: Merged Dataset

```

185 proc sql;
186 create table all_total as
187 select "Total" as provider_name, sum(all_id) as all_id, sum(consent_ID) as consent_ID,
188 sum(isel16_ID) as isel16_id, sum(phi1_ID) as phi1_ID,
189 calculated consent_ID / calculated all_ID as pct_yes format percent13.2,
190 calculated isel16_ID / calculated consent_ID as pct_complete format percent13.2,
191 calculated phi1_ID / calculated consent_ID as pct_primary format percent13.2
192 from all2 ;
193 quit ;
194
195 data all3 ;
196 set all2 all_total ;
197 run;
198
199 *** add a subtotal per ARM ***;
200 data all3 ;
201 set all3 ;
202 length arm2 $10 ;
203 if provider_id in (1000137, 1000191, 1000365, 1000701, 1000190) then arm2 = 'Prov' ;
204 else if provider_id in (1000961, 1001046, 1000811, 1000468, 1000491, 1000516) then arm2 = 'Person' ;
205 else arm2 = 'Total' ;
206 run;

```

An overall summary record with totals and percentages is created from the table shown in Display 9.

Once stacked to the provider summary dataset, a new grouping variable is created categorizing each provider by the treatment arm.

Display 10: Overall Summary Statistics and Treatment Arm Variable Creation

	provider_name	all_ID	ConsentTF	consent_ID	isel16TF	isel16_ID	phiTF	phi1_ID	pct_yes	pct_complete	pct_primary	arm2
1	Provider 1	120	1	54	1	54	1	54	45.00%	100.00%	100.00%	Person
2	Provider 2	176	1	82	1	79	1	80	46.59%	96.34%	97.56%	Person
3	Provider 3	140	1	83	1	83	1	83	59.29%	100.00%	100.00%	Prov
4	Provider 4	231	1	113	1	113	1	113	48.92%	100.00%	100.00%	Prov
5	Provider 5	74	1	27	1	27	1	27	36.49%	100.00%	100.00%	Prov
6	Provider 6	96	1	62	1	61	1	61	64.58%	98.39%	98.39%	Person
7	Provider 7	102	1	28	1	28	1	28	27.45%	100.00%	100.00%	Person
8	Provider 8	445	1	306	1	306	1	306	68.76%	100.00%	100.00%	Prov
9	Provider 9	133	1	99	1	99	1	99	74.44%	100.00%	100.00%	Person
10	Provider 10	205	1	175	1	174	1	175	85.37%	99.43%	100.00%	Prov
11	Provider 11	126	1	70	1	70	1	70	55.56%	100.00%	100.00%	Person
12	Total	1848	.	1099	.	1094	.	1096	59.47%	99.55%	99.73%	Total

Display 11: Dataset with Overall Summary Statistics

```

208 proc sql;
209 create table arm_total as
210 select arm2, sum(all_id) as all_id, sum(consent_ID) as consent_ID,
211 sum(isel16_ID) as isel16_id, sum(phi1_ID) as phi1_ID,
212 calculated consent_ID / calculated all_ID as pct_yes format percent13.2,
213 calculated isel16_ID / calculated consent_ID as pct_complete format percent13.2,
214 calculated phi1_ID / calculated consent_ID as pct_primary format percent13.2
215 from all3
216 where arm2 ne 'Total'
217 group by arm2 ;
218 quit ;
219
220 data all4 ;
221 set all3 arm_total ;
222 if arm2 = 'Person' and provider_ID = . then provider_name = 'Person Directed ARM'; else
223 if arm2 = 'Prov' and provider_ID = . then provider_name = 'Provider Directed ARM';
224 run;
225
226 proc sort data = all4 ; by arm2 ; run;

```

Summary totals and percentages by treatment arm are created.

This completes the report dataset.

Display 12: Final Output Table Creation

VIEWTABLE: Work.All4													
	provider_name	all_ID	ConsentTF	consent_ID	isel16TF	isel16_ID	phiTF	phi1_ID	pct_yes	pct_complete	pct_primary	arm	
1	Provider 1	120	1	54	1	54	1	54	45.00%	100.00%	100.00%	Person	
2	Provider 2	176	1	82	1	79	1	80	46.59%	96.34%	97.56%	Person	
3	Provider 3	96	1	62	1	61	1	61	64.58%	98.39%	98.39%	Person	
4	Provider 4	102	1	28	1	28	1	28	27.45%	100.00%	100.00%	Person	
5	Provider 5	133	1	99	1	99	1	99	74.44%	100.00%	100.00%	Person	
6	Provider 6	126	1	70	1	70	1	70	55.56%	100.00%	100.00%	Person	
7	Person Directed ARM	753	.	395	.	391	.	392	52.46%	98.99%	99.24%	Person	
8	Provider 7	140	1	83	1	83	1	83	59.29%	100.00%	100.00%	Prov	
9	Provider 8	231	1	113	1	113	1	113	48.92%	100.00%	100.00%	Prov	
10	Provider 9	74	1	27	1	27	1	27	36.49%	100.00%	100.00%	Prov	
11	Provider 10	445	1	306	1	306	1	306	68.76%	100.00%	100.00%	Prov	
12	Provider 11	205	1	175	1	174	1	175	85.37%	99.43%	100.00%	Prov	
13	Provider Directed ARM	1095	.	704	.	703	.	704	64.29%	99.86%	100.00%	Prov	
14	Total	1848	.	1099	.	1094	.	1096	59.47%	99.55%	99.73%	Total	

Display 13: Final Output Dataset

The final report was produced using PROC REPORT in combination with ODS PDF. The code is shown in Display 14 and Display 15.

```

232 ods escapechar = "&" ;
233 Options orientation=landscape nonumber ;
234
235 ods pdf
236 body = "\\Client\C$\Users\parthasarathym\Documents\PCORI_Share\enroll_status_&sysdate9..pdf"
237 style = meadow ;
238
239 Title " ";
240 Title2 "PCORI Provider Weekly Member Enrollment Status Report ";
241 Title3 " ";
242 Title4 " ";
243 title5 "Report date: &sysdate9 ";
244 title6 " ";
245 footnote height=8pt "Community Care Behavioral Health | Department:DS/DA " ;
246
247 proc report data = all4 missing nowindows headline headskip split="*" nowd
248 style(report)=(cellpadding=3pt) ;
249
250 column provider_name provider_id all_id consentTF consent_ID pct_yes
251 phiTF phi1_ID pct_primary isel16TF isel16_ID pct_complete arm2 arm ;
252
253 define provider_name / display "Provider Name"
254 style={cellwidth=25% just=left font_face=calibri font_size=2} ;
255
256 define provider_id / display "Provider ID"
257 style={just=center font_face=calibri font_size=2} ;
258
259 define arm2 / noprint ;
260 define arm / noprint " " center ;

```

ODS PDF was used to format the look and feel of the report including adding custom titles and footnotes.

SAS System options used to orient the report and date stamp and time stamp the output.

PROC REPORT was used to customize the look and feel of the table with columns being defined in specific format and suppressing some columns as needed.

Display 14: PROC REPORT for Final Output Creation (part 1)

```

262 compute arm ;
263 if provider_Id in (1000137, 1000191, 1000365, 1000701, 1000190) then do;
264   call define(_row_, "style", "style=[background=#CCFF99]");
265 end;
266 else if provider_Id in (1000961, 1001046, 1000811, 1000468, 1000491, 1000516) then do ;
267   call define(_row_, "style", "style=[background=#CCFFFF]");
268 end;
269 else if provider_name in ('Person Directed ARM', 'Provider Directed ARM') then do ;
270   call define(_row_, "style", "style=[background=#FF8B3B]");
271 end;
272 endcomp;
273
274 define all_id / display "Members Identified"
275 style={cellwidth=8% just=center font_face=calibri font_size=2};
276
277 define consentTF / noprint ;
278
279 define consent_id / display "Members with Consent"
280 style={cellwidth=8% just=center font_face=calibri font_size=2};
281
282 define pct_yes / display "Consent Rate" format=percent13.2
283 style={cellwidth=8% just=center font_face=calibri font_size=2};
284
285 define phiTF / noprint ;
286
287 define phi1_ID / display "Members who completed Primary Measures"
288 style={cellwidth=8% just=center font_face=calibri font_size=2};
289
290 define pct_primary / display "Primary Measure Rate" format=percent13.2
291 style={cellwidth=8% just=center font_face=calibri font_size=2};
292
293 define isel16TF / noprint ;
294
295 define isel16_id / display "Members who Completed all Measures"
296 style={cellwidth=8% just=center font_face=calibri font_size=2};
297
298 define pct_complete / display "Completion Rate" format=percent13.2
299 style={cellwidth=8% just=center font_face=calibri font_size=2};
300
301 compute after_page_ / Left
302   style=[just=left font_face=calibri font_size=2 font_weight=bold];
303   Line "{dagger}Rates are cumulative starting from 9/18/2013. ~n
304   ~{dagger}Denominator for Consent Rate is members identified at each provider site. ~n
305   ~{dagger}Denominator for Primary Measure Rate & Completion Rate is members who consented for PCORI at each provider site. ~n
306   ~{dagger}Provider Supported Arm = Green; Person Directed Arm = Blue. ~n
307   ~{dagger}PCORI Treatment Sub Totals = Orange. "
308 ;
309 endcomp;
310 run;
311 OBS pdf CLOSE;
312 OBS LISTING;

```

One highly useful utility of PROC REPORT was the COMPUTE block which in this instance was used to traffic light the report table based on project colors and also to highlight the treatment arm summary record and the overall summary record.

Another very useful feature was the COMPUTE AFTER block which could be used to explain the numbers being presented in the report in a very concise format.

Display 15: PROC REPORT for Final Output Creation (part 2)

The final report output sent to all providers and project stakeholders is shown in Output 1.

PCORI Provider Weekly Member Enrollment Status Report								09:18 Thursday, February 20, 2014
Report date: 20FEB2014								
Provider Name	Provider ID	Members Identified	Members with Consent	Consent Rate	Members who completed Primary Measures	Primary Measure Rate	Members who Completed all Measures	Completion Rate
Provider 1	1000961	120	54	45.00%	54	100.00%	54	100.00%
Provider 2	1001046	176	82	46.59%	80	97.56%	79	96.34%
Provider 3	1000811	96	62	64.58%	61	98.39%	61	98.39%
Provider 4	1000468	102	28	27.45%	28	100.00%	28	100.00%
Provider 5	1000491	133	99	74.44%	99	100.00%	99	100.00%
Provider 6	1000516	126	70	55.56%	70	100.00%	70	100.00%
Person Directed ARM		753	395	52.46%	392	99.24%	391	98.99%
Provider 7	1000137	140	83	59.29%	83	100.00%	83	100.00%
Provider 8	1000191	231	113	48.92%	113	100.00%	113	100.00%
Provider 9	1000365	74	27	36.49%	27	100.00%	27	100.00%
Provider 10	1000701	445	306	68.76%	306	100.00%	306	100.00%
Provider 11	1000190	205	175	85.37%	175	100.00%	174	99.43%
Provider Directed ARM		1095	704	64.29%	704	100.00%	703	99.86%
Total		1848	1099	59.47%	1096	99.73%	1094	99.55%

†Rates are cumulative starting from 9/18/2013.
†Denominator for Consent Rate is members identified at each provider site.
†Denominator for Primary Measure Rate & Completion Rate is members who consented for PCORI at each provider site.
†Provider Supported Arm = Green; Person Directed Arm = Blue.
†PCORI Treatment Sub Totals = Orange.

Community Care Behavioral Health | Department:DS/DA

Output 1: Final Report Output

CONCLUSION

A variety of SAS® functions and procedures have proven to be quite useful in addressing the multiple requirements of the PCORI project. As the project moves forward, we will continue to look to SAS solutions to solve our technical challenges.

REFERENCES

- United States Government Printing Office. "Public Law 111 - 148 - Patient Protection and Affordable Care Act." Available at <http://www.gpo.gov/fdsys/pkg/PLAW-111publ148/content-detail.html>.
- Patient Centered Outcomes Research Institute. "Mission and Vision." Available at <http://www.pcori.org/about-us/mission-and-vision/>.
- National Institute of Mental Health. "The numbers count: Mental disorders in America." Available at www.nimh.nih.gov/publicat/numbers.cfm.
- Dembling BP, Chen DT, Vachon L. 1999. "Life expectancy and causes of death in a population treated for serious mental illness." Psychiat Serv. 1036-1042.

ACKNOWLEDGMENTS

The authors would like to acknowledge Jane Kogan and the Center for High Value Healthcare at UPMC for their assistance in applying for and obtaining the PCORI funding and their continued guidance through the life of the project. The authors would also like to acknowledge the efforts of Robin Allison in researching methods for extracting data from excel into SAS directly which helped this project immensely and Elizabeth Venturino for her efforts in creating custom Cognos report queries that extracted PCORI data from the web portal and put it into excel in a format that was easy to transfer to SAS.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: Michele Mesiano
Organization: Community Care Behavioral Health Organization
Address: One Chatham Center, Suite 700
City, State ZIP: Pittsburgh, PA 15219
Work Phone: 412-402-7519
Email: mesianomm@ccbh.com

Name: Meghna Parthasarathy
Organization: Community Care Behavioral Health Organization
Address: One Chatham Center, Suite 700
City, State ZIP: Pittsburgh, PA 15219
Work Phone: 412-402-7507
Email: parthasarathym@ccbh.com

Name: Lauren Terhorst, Ph.D.
Organization: University of Pittsburgh
Address: 5021 Forbes Tower
City, State ZIP: Pittsburgh, PA 15260
Work Phone: 412-383-1141
Email: lat15@pitt.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.